SECTION 2 : Algebra

In the first section we saw that Maple could be used as a simple calculator to add, subtract, multiply, divide, take powers etc. and more. In this section we will explore Maple's ability to solve equations and manipulate algebraic expressions.

2.1 ALGEBRAIC EXPRESSIONS AND SUBSTITUTION

PARTIAL FRACTIONS

Maple can work out a partial fraction decomposition of a given polynomial quotient for you using the following form of '*convert*'.

```
> # Decompose the following:
> q:=(x^2 + 1)/(x^3 - 7*x + 6);
```

$$q := \frac{x^2+1}{x^3-7\,x+6}$$

```
> convert(q,parfrac,x);
```

$$-\frac{1}{2\,(x-1)}+\frac{1}{x-2}+\frac{1}{2\,(x+3)}$$

```
> # In the command the final entry tells Maple the variable in the
  problem (in this case 'x').
```

THE *SUBS* COMMAND

The command *subs(x=a,expr)* substitutes the value *a* in place of *x* in expression '*expr*'. For example:

```
> # Evaluate x^2-4*x+3 when x=2.5 using subs
> subs(x=2.5,x^2-4*x+3);
```

$$-0.75$$

```
> # To substitute more than one variable in an expression place
  curly brackets around the items to be substituted.
> # Substitute x=r*cos(theta) and y=r*sin(theta) into x^2+y^2, and
  simplify
> subs({x=r*cos(theta), y=r*sin(theta)},x^2+y^2);
```

$$r^2\cos(\theta)^2+r^2\sin(\theta)^2$$

```
> simplify(%);
```

$$r^2$$

subs can also be used to double check that Maple has correctly identified the solution to a problem.

```
> # Find the solutions of the quadratic equation x^2-5*x+6=0, and
  confirm the solutions by substitution
> y1:=x^2-5*x+6; sols:=solve(y1,x);
```

$$y1 := x^2-5\,x+6$$

$$sols := 3, 2$$

```
> subs(x=sols[1],y1);
```

$$0$$

```
> subs(x=sols[2],y1);
```

$$0$$

```
> # Thus x=3 and x=2 are indeed both solutions of the problem
```

2.2 SOLVING EQUATIONS AND SYSTEMS OF EQUATIONS

We saw in section 1 how to use *solve* to find the solution to polynomial equations. We can also use *solve* for dealing with more general algebraic equations.

```
> # Given that 2/a -3/b =4 solve for b in terms of a. Then use subs
  to determine the specific solution for b when a=4
> s1:=solve(2/a-3/b=4,b);  # It is clear from this example why we
  must identify the variable we are solving for in the solve
  command!
```

$$s1 := -\frac{3\,a}{2\,(-1+2\,a)}$$

```
> subs(a=4,s1);
```

$$\frac{-6}{7}$$

```
> # Example: solve x^2+x=y^2-y for x in terms of y. Show there is
  only one distinct solution when y=1/2.
> sols:=solve(x^2+x=y^2-y,x);
```

$$sols := -y, -1+y$$

```
> subs(y=1/2,sols[1]); subs(y=1/2,sols[2]);
```

$$\frac{-1}{2}$$

$$\frac{-1}{2}$$

```
> # so the two solutions are identical when y=1/2.
>
```

We can also use *solve* with a system of equations. In this case we should enclose all of the equations in curly brackets, and all of the variables in another set of curly brackets. For example:

```
> # Find the solution of the set of the simultaneous equations
  x+2*y=3, 4*x-7*y=2
> solve({x+2*y=3, 4*x-7*y=2},{x,y});
```

$$\{y=\frac{2}{3}, x=\frac{5}{3}\}$$

```
> # Find the solution of the set of equations 4*u-2*v+w=1, 3*w+2*u=v
  and u+v+w=100
> eq1:=4*u-2*v+w=1; eq2:= 3*w+2*u=v; eq3:=u+v+w=100;  # Easier to
  assign the equations first in this case
```

$$eq1 := 4\,u-2\,v+w=1$$

$$eq2 := 3\,w+2\,u=v$$

$$eq3 := u + v + w = 100$$

```
> solve({eq1,eq2,eq3},{u,v,w});
```

$$\{u = \frac{168}{5}, w = \frac{-1}{5}, v = \frac{333}{5}\}$$

USING *FSOLVE* FOR FINDING APPROXIMATE SOLUTIONS OF EQUATIONS

In many cases we cannot find an exact solution for an algebraic equation. However we can numerically approximate the solution using the Maple command *fsolve* - this command does not always find all solutions without a little help. You should always use a plot to ensure all solutions have been found!!

```
> # Find the solutions of sin(x)-x/2=0
> solve(sin(x)-x/2=0,x);
```

$$\mathrm{RootOf}(-\_Z + 2\sin(\_Z))$$

```
> # Maple is just restating the problem. This indicates that solve
  is not going to give us a solution. Plot the function and try
  fsolve instead
> plot(sin(x)-x/2,x=-3*Pi..3*Pi);
```




```
> # From the plot we see there are three solutions
> fsolve(sin(x)-x/2=0,x);
```

$$0.$$

Here *fsolve* has only found one of the solutions (the easiest, $x=0$). We must help Maple to locate the others by including the intervals in which the roots lie.

```
> fsolve(sin(x)-x/2=0,x,-3..-1);  # Only look for a solution in the
  interval (-3,-1).
```

-1.895494267

```
> fsolve(sin(x)-x/2=0,x,1..3);
```

1.895494267

```
> # Find all solutions of cos(x)=3*x
> solve(cos(x)=3*x,x);  # try solving exactly
```

$\mathrm{RootOf}(-3\ \_Z + \cos(\_Z))$

```
> # can't find an exact solution so plot the function and use fsolve
> plot({cos(x),3*x},x=-2*Pi..2*Pi);
```




```
> # from the plot we see there is a single solution (where the two
  curves cross).
> fsolve(cos(x)=3*x,x);
```

0.3167508288

```
> # which is found directly by fsolve.
```

2.3 MATRICES AND LINEAR ALGEBRA

Maple has lots of built in commands for dealing with linear algebra - we will only get a chance to look at a few of them here. For further information see the online help and books in the library.
IMPORTANT : For most of the commands in this section you need to load in the linear algebra package by typing
with(linalg):
at the command prompt (using a semicolon instead of a colon lists all of the commands available in the linalg package).

```
> with(linalg):
Warning, the protected names norm and trace have been redefined and unprotected
> # Matrices are input using the matrix command. There are two ways
  of entering a matrix, either by enclosing the entries of each row
```

```
in a set of square brackets:
> A:=matrix([[1,2,3],[4,5,6]]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix}$$

```
> # or telling Maple the dimensions of the matrix (number of rows
  followed by number of columns) and including all entries in a
  single set of square brackets:
> B:=matrix(3,2,[3,0,4,6,2,1]);
```

$$B := \begin{bmatrix} 3 & 0 \\ 4 & 6 \\ 2 & 1 \end{bmatrix}$$

MATRIX MULTIPLICATION is achieved using &* rather than just *.

```
> A&*B;
```

$$A \mathbin{\&*} B$$

```
> # Maple needs some encouragement to show the result, this can be
  achieved using the 'evalm' command (evaluate a matrix expression).
  For example:
> evalm(A&*B);
```

$$\begin{bmatrix} 17 & 15 \\ 44 & 36 \end{bmatrix}$$

```
> # Similarly find the product of B times A
> evalm(B&*A);
```

$$\begin{bmatrix} 3 & 6 & 9 \\ 28 & 38 & 48 \\ 6 & 9 & 12 \end{bmatrix}$$

```
> # Make sure your matrices have the correct dimensions for
  multiplying or you will get an error
> evalm(A&*A);
Error, (in linalg[multiply]) non matching dimensions for vector/matrix product
> 
```

MATRIX OPERATIONS: determinants, transpose, inverse

Standard matrix operations can be carried out easily in Maple. For example

```
> A:=matrix(3,3,[4,7,2,1,3,4,3,9,1]);  # a square matrix
```

$$A := \begin{bmatrix} 4 & 7 & 2 \\ 1 & 3 & 4 \\ 3 & 9 & 1 \end{bmatrix}$$

```
> # Find the transpose and determinant of A
> At:=transpose(A);
```

$$At := \begin{bmatrix} 4 & 1 & 3 \\ 7 & 3 & 9 \\ 2 & 4 & 1 \end{bmatrix}$$

```
> det(A);
```

$$-55$$

```
> # the determinant is non-zero, so lets try to find the inverse of A
> inA:=inverse(A);
```

$$inA := \begin{bmatrix} \frac{3}{5} & \frac{-1}{5} & \frac{-2}{5} \\ \frac{-1}{5} & \frac{2}{55} & \frac{14}{55} \\ 0 & \frac{3}{11} & \frac{-1}{11} \end{bmatrix}$$

```
> evalm(inA&*A); # to convince ourselves this is the inverse
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

SOLVING LINEAR EQUATIONS

There are several ways to solve equations of the form $Ax=b$ where x and b are vectors

```
> # Use A from above and let b be the column vector (1 2 3), find the solution of Ax=b using the matrix inverse
> b:=matrix(3,1,[1,2,3]);
```

$$b := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

```
> x:=evalm(inverse(A)&*b);
```

$$x := \begin{bmatrix} -1 \\ \frac{7}{11} \\ \frac{3}{11} \end{bmatrix}$$

```
> # In many cases the command linsolve is better as it can still find a solution when the determinant of the matrix is zero.
> linsolve(A,b);
```

$$\begin{bmatrix} -1 \\ \frac{7}{11} \\ \frac{3}{11} \end{bmatrix}$$

```
> # The same solution as above.
```

2.4 THE 'ARROW' NOTATION FOR DEFINING FUNCTIONS

We can use an arrow -> made up of a minus sign and a greater than symbol to represent a function in Maple. for example

f:=x->x^2

associates *f* with the function that squares its argument - the variable *x* is a dummy:

```
> x:='x';y:='y';z:='z';  # unassign x y and z before we start
```

$$x := x$$

$$y := y$$

$$z := z$$

```
> f:=x->x^2;
```

$$f := x \to x^2$$

```
> f(5); f(-2.5); f(y); f(z);  # evaluate the function with different
  arguments
```

$$25$$

$$6.25$$

$$y^2$$

$$z^2$$

```
> # We can also compose functions:
> g:=x->sqrt(x-1);
```

$$g := x \to \sqrt{x-1}$$

```
> f(g(x));
```

$$x - 1$$

```
> g(f(x));
```

$$\sqrt{x^2 - 1}$$

```
> # clearly demonstrating that f(g(x)) is not the same as g(f(x))
```